Effect of Significant Digits in Solution of Simultaneous Linear Equations

©2006 Kevin Martin, Autar Kaw, Jamie Trahan

University of South Florida

United States of America

kaw@eng.usf.edu

NOTE: This worksheet demonstrates the use of Maple to illustrate the effect of significant digits in the solution of simultaneous linear equations.

```
> restart;
```

Introduction

The number of significant digits used in numerical solutions of simultaneous linear equations influences the accuracy of the solution vector, especially if the coefficient matrix is nearly singular. In this worksheet, the reader can choose a system of equations and see the influence of significant digits on each element of the solution vector. To learn more about the effects of significant digits on the accuracy of the solution vector click here.

The following simulation uses Naïve Gaussian Elimination method to demonstrate the effect that significant digits have on the accuracy of the solution.

Section 1: Input Data

The following are the input parameters to begin the simulation. This is the only section that requires user input. In the simulation, Naïve Gaussian Elimination method is used to solve a set of simultaneous linear equations, [A] [X] = [RHS], where $[A]_{nxn}$ is the square coefficient matrix, $[X]_{nx1}$ is the solution vector, and $[RHS]_{nx1}$ is the right hand side array. To demonstrate the effect that significant digits have on the accuracy of solution, Maple will return a list of solution vectors that were calculated using the significant digits within the range of your choice. It will then plot each element of the solution vector as a function of the number of significant digits used.

Input parameters:
n = number of equations
[AL] = nxn coefficient matrix
[RHS] = nx1 right hand side array

lower_lim = lower limit of range of significant digits
upper_lim = upper limit of range of significant digits

```
> restart;
  n:=3;
  AL:=Matrix([[12.,7.,3.],[1.,5.,1.],[13.,12.,4.001]]);
  RHS:=[22.,7.,29.001];
  lower_lim:=3;
  upper_lim:=10;
```

$$n := 3$$

$$AL := \begin{bmatrix} 12. & 7. & 3. \\ 1. & 5. & 1. \\ 13. & 12. & 4.001 \end{bmatrix}$$

$$RHS := [22., 7., 29.001]$$

$$lower\_lim := 3$$

$$upper\_lim := 10 \quad (2.1)$$

Section 2: Procedure for Naïve Gaussian Elimination Method

Parameter names:
n = number of equations
[A] = *n*x*n* coefficient matrix
[B] = *n*x1 right hand side vector
dig = number of significant digits used in calculations

```
> gauss_Naïve:=proc(n,A,B,dig)
  local k,i,multiplier,j,sum,X,AA,BB:
  #Assigning the number of significant digits to be used in the calculations.
  Digits:=dig;
  #Assigning the [A] and [B] matrices to different variables because input variables should not be placed on the LHS of an assignment in a procedure.
  for i from 1 by 1 to n do
     for j from 1 by 1 to n do
       AA[i,j]:=A[i,j];
     end do:
  BB[i]:=B[i];
  end do;
```

```
X:=Array(1..n);

#Forward Elimination Steps:
#Conducting (n-1) steps of forward elimination.
for k from 1 by 1 to n-1 do
    #Using elementary row operations to transform the coefficient matrix into upper triangular form.
    for i from (k+1) by 1 to n do
     #Calculating the value that is multiplied to each equation.
     multiplier:=AA[i,k]/AA[k,k]:
        for j from (k+1) by 1 to n do
            #Reducing the number of unknowns in the jth row.
            AA[i,j]:=AA[i,j]-multiplier*AA[k,j]:
        end do:
     #Conducting forward elimination steps on the [RHS] vector.
     BB[i]:=BB[i]-multiplier*BB[k]:
  end do:
end do:

#Back Substitution Steps:
#Solving for the nth equation as it has only one unknown.
X[n]:=BB[n]/AA[n,n]:
#Solving for the remaining (n-1) unknowns working backwards from the (n-1)th equation to the first equation.
for i from (n-1) by (-1) to 1 do
    #Initializing the series sum to zero.
    sum:=0:
    for j from (i+1) to n by 1 do
        #Substituting known values into the ith equation.
        sum:=sum+AA[i,j]*X[j]:
    end do:
#Solving for the ith unknown.
X[i]:=(BB[i]-sum)/AA[i,i]:
end do:
return(X):
end proc:
```

Section 3: Results

In this section, the procedure for Naïve Gaussian Elimination is called for different numbers of significant digits within the lower_lim to upper_lim significant digit range. Each [X] solution vector is indexed according to the number of significant digits used.

```
> X:=Array(1..upper_lim):
  for k from lower_lim by 1 to upper_lim do
  X[k]:=gauss_Naïve(n,AL,RHS,k);
  end do;
```

$$X_3 := [0.700 \quad 0.661 \quad 3.00]$$

$$X_4 := [0.9242 \quad 0.9151 \quad 1.500]$$

$$X_5 := [0.97258 \quad 0.96914 \quad 1.1818]$$

$$X_6 := [0.997008 \quad 0.996638 \quad 1.01980]$$

$$X_7 := [0.9996983 \quad 0.9996606 \quad 1.001998]$$

$$X_8 := [0.99996983 \quad 0.99996604 \quad 1.0002000]$$

$$X_9 := [0.999996983 \quad 0.999996604 \quad 1.00002000]$$

$$X_{10} := [0.9999996983 \quad 0.9999996604 \quad 1.000002000] \tag{4.1}$$

The exact solution to the system of equations with default number of significant digits in Maple is:

```
> with(linalg):
  Xexact:=linsolve(AL,RHS);
Warning, the protected names norm and trace have been redefined
and unprotected
```

$$Xexact := [0.9999998948 \quad 0.9999998811 \quad 1.000000700] \tag{4.2}$$

The following graphs also demonstrate the effect that the number of significant digits has on the solution. Each element of the solution vector is plotted as a function of the number of significant digits used.

```
> 'X':
  for k from lower_lim by 1 to upper_lim do
  X:=gauss_Naïve(n,AL,RHS,k):
     for i from 1 by 1 to n do
        Xstore[i,k]:=X[i]:
  end do:
  end do:
  with(plots):
  for i from 1 by 1 to n do
  ttl:=cat(`Value of X`,i,` as a function of the number of
  significant digits used`):
  lbl:=cat(`x`,i):
  data:=[seq([k,Xstore[i,k]],k=lower_lim..upper_lim)];
  pointplot(data,connect=true,color=blue,axes=boxed,title=ttl,
  titlefont=[HELVETICA,25,BOLD],axes=BOXED,labels=["number of
```

```
significant digits",lbl],labelfont=[HELVETICA,12,BOLD],
thickness=2);
end do;
```

Warning, the name changecoords has been redefined

$ttl := Value\ of\ X1\ as\ a\ function\ of\ the\ number\ of\ significant\ digits\ used$

$lbl := x1$

$data := [[3, 0.700], [4, 0.9242], [5, 0.97258], [6, 0.997008], [7, 0.9996983], [8, 0.99996983], [9, 0.999996983], [10, 0.9999996983]]$



$ttl := Value\ of\ X2\ as\ a\ function\ of\ the\ number\ of\ significant\ digits\ used$

$lbl := x2$

$data := [[3, 0.661], [4, 0.9151], [5, 0.96914], [6, 0.996638], [7, 0.9996606], [8, 0.99996604], [9, 0.999996604], [10, 0.9999996604]]$



ttl := *Value of X3 as a function of the number of significant digits used*

lbl := *x3*

data := [[3, 3.00], [4, 1.500], [5, 1.1818], [6, 1.01980], [7, 1.001998], [8, 1.0002000], [9, 1.00002000], [10, 1.000002000]]



References

[1] Autar Kaw*, Holistic Numerical Methods Institute, http://numericalmethods.eng.usf.edu/mws,* See
Introduction to Systems of Equations
Effect of Significant Digits on Solution of Equations
How does Gaussian Elimination Work?

Conclusion

Maple helped us to apply our knowledge of Naïve Gaussian Elimination to study the effect of significant digits on the solution of a set of simultaneous linear equations.

Question 1: Choose a set of equations for which the coefficient matrix is nonsingular. For example

$$5x + 6y + 9z = 29$$
$$6x + 9y + 2z = 19$$

$$11x + 9y + 5z = 30$$

See how the number of significant digits makes a difference in the solution vector.

Question 2: Choose a set of equations for which the coefficient matrix is nearly singular. For example

$$5x + 6y + 9z = 29$$
$$6x + 9y + 2z = 19$$
$$11x + 15y + 11.001z = 49.002$$

See if the number of significant digits makes a difference in the solution vector.

Question 3: One of the classical problems to show the effect of significant digits on solutions of simultaneous linear equations is with a Hilbert matrix as the coefficient matrix. A matrix $[H]_{nxn}$ is called the n^{th} Hilbert matrix if

$$h_{ij} = \frac{1}{i + j - 1}$$

For example, a 4x4 Hilbert matrix is

$$Matrix\left(\left[\left[1, \frac{1}{2}, \frac{1}{3}\right], \left[\frac{1}{2}, \frac{1}{3}, \frac{1}{4}\right], \left[\frac{1}{3}, \frac{1}{4}, \frac{1}{5}\right]\right]\right)$$

$$\begin{bmatrix} 1 & \frac{1}{2} & \frac{1}{3} \\ \frac{1}{2} & \frac{1}{3} & \frac{1}{4} \\ \frac{1}{3} & \frac{1}{4} & \frac{1}{5} \end{bmatrix} \tag{6.1}$$